

Data Analysis & Business Intelligence Made Easy with Excel Power Tools
Excel Data Analysis Basics = E-DAB
Notes for Video:
E-DAB-03: Summary Reports with Excel Spreadsheet Formulas

Objectives of Video:

1. Excel Spreadsheet Formulas Uses in Data Analysis	2
2. Change Default Settings for Table Formula Nomenclature (Structured References).	3
3. Cell References in Formulas.....	4
4. Excel Functions to Make Calculations Based on one Condition, four examples:	4
5. SUMIFS Function with One Condition shown on next page in cell H13	4
6. SUMIFS Function with One Condition & Dynamic Array shown on next page in cell K13.....	4
7. COUNTIFS Function with One Condition shown on next page in cell H24	4
8. COUNTIFS Function with One Condition & Dynamic Array shown on next page in cell K24	4
9. Excel Functions to Make Calculations Based on an AND Logical Text.	6
10. SUMIFS for Summary Report Based on Adding, Old School Cell References and Dynamic Arrays.	7
11. COUNTIFS for Summary Report Based on Counting, Old School Cell References and Dynamic Arrays.....	7
12. AVERAGEIFS for Summary Report Based on Averaging, Old School Cell References and Dynamic Arrays.....	7
13. Screen Shots of Old School Formulas with Totals on Worksheet Named “Old School with Totals”.	8
14. Office 365 Dynamic Array Notes from Worksheet named “O365 Dynamic Arrays”	10

1. Excel Spreadsheet Formulas Uses in Data Analysis

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		Objective #1: Excel Spreadsheet Formulas Uses in Data Analysis											
3													
4		Excel Spreadsheet Formulas are the only feature in Excel that update instantly when source data changes.											
5		When you want a Data Analysis Excel Solution to respond instantly to changes in the source data, use formulas.											
6		Other Data Analysis Excel Solution such as PivotTables and Power Query respond to changes in the source data only after using the refresh command.											
7													
8		Date	Sales Rep	Autos	Sales		Total Sales (\$)						
9		2/11/2019	Alma	Chevy	\$38,016		Autos/ Sales Rep	Alma	Jaeyoung	Tyrone	Total		
10		2/11/2019	Tyrone	Chevy	\$71,182		Chevy	107,614	57,034	71,182	235,830		
11		2/11/2019	Jaeyoung	Honda	\$70,720		Ford	37,432	63,092	98,481	199,005		
12		2/11/2019	Alma	Chevy	\$69,598		Honda	33,415	70,720	213,010	317,145		
13		2/11/2019	Tyrone	Honda	\$55,299		Toyota	47,352	95,940	0	143,292		
14		2/11/2019	Jaeyoung	Toyota	\$63,882		Total	225,813	286,786	382,673	895,272		
15		2/12/2019	Alma	Toyota	\$47,352								
16		2/12/2019	Tyrone	Ford	\$53,105								
17		2/12/2019	Jaeyoung	Toyota	\$32,058		Sum of Sales (\$)	Sales Rep					
18		2/12/2019	Alma	Honda	\$33,415		Autos	Alma	Jaeyoung	Tyrone	Grand Total		
19		2/12/2019	Tyrone	Honda	\$71,880		Chevy	107,614	57,034	71,182	235,830		
20		2/13/2019	Jaeyoung	Ford	\$63,092		Ford	37,432	63,092	98,481	199,005		
21		2/13/2019	Alma	Ford	\$37,432		Honda	33,415	70,720	213,010	317,145		
22		2/13/2019	Tyrone	Ford	\$45,376		Toyota	47,352	95,940		143,292		
23		2/13/2019	Jaeyoung	Chevy	\$57,034		Grand Total	225,813	286,786	382,673	895,272		
24		2/13/2019	Tyrone	Honda	\$45,881								
25		2/13/2019	Tyrone	Honda	\$39,950								
26													
27													
28		1/14/2019	Alma	Ford	31980								
29		1/14/2019	Tyrone	Toyota	45789								
30		1/14/2019	Jaeyoung	Honda	55299								
31		1/14/2019	Alma	Toyota	63882								
32		1/14/2019	Tyrone	Toyota	47352								
33		1/14/2019	Jaeyoung	Chevy	29870								
34		1/14/2019	Alma	Honda	31852								

Reports built with **Excel Spreadsheet Formulas** update instantly when source data changes.

Reports built with **Excel PivotTables** require you to refresh the Report when source data changes.

As seen in the video, when this new data is added to the Excel Table, the **formula report will respond instantly.**

2. Change Default Settings for Table Formula Nomenclature (Structured References).

When we refer to a Field or Column in an Excel Table, we can choose one of two options:


1. Table Name and then field name in square brackets, called Table Formula Nomenclature (or Structured References), as seen here:

Autos	Sales	Total Sales (\$)			
Chevy	\$38,016	Autos/ Sales	Alma	Jaeyoung	Tyrone
Chevy	\$71,182	Chevy	=SUMIFS(fSales[Sales]		
Honda	\$70,720	Ford	SUMIFS(sum_range, criteria_range1, criteria1, ...)		
Chevy	\$69,598	Honda			

2. We can use the actual cell references as seen here:

Autos	Sales	Total Sales (\$)			
Chevy	\$38,016	Autos/ Sales	Alma	Jaeyoung	Tyrone
Chevy	\$71,182	Chevy	=SUMIFS(\$E\$12:\$E\$28		
Honda	\$70,720	Ford	SUMIFS(sum_range, criteria_range1, criteria1, ...)		
Chevy	\$69,598	Honda			

3. To change the default setting for referring to a File or Column in an Excel Table, Go to the File Ribbon Tab, Then to Options, then on the left click on the Formula tab, then check or uncheck the option "", as seen here:



3. Cell References in Formulas.

1) Example of Cell Reference: A1

- i. Column reference = A
- ii. Row reference = 1

2) Copying formulas with Cell References:

- i. When we copy a formula that contains cell references, we need to consider whether we need: Relative, Absolute, Mixed with the Column Locked or Mixed with the Row Locked.
- ii. If you will not copy the formula, there is no need to consider what type of cell reference it will be.

3) Four Basic Types of Cell References (Relative, Absolute, Mixed Column Locked, Mixed Row Locked):

i. Relative Cell References – Example: A1

- No dollar signs
- Moves relatively throughout the copy action.
- Relatively means that if the formula is looking at a cell reference that is three cells to the left, when you copy the formula to any other cell, the cell reference will still be looking three cells to the left.

ii. Absolute Cell References – Example: \$A\$1

- Dollar signs before both:
 - i. Column reference = A
 - ii. Row reference = 1
- Absolute means that if the formula is looking at a particular cell reference, when you copy the formula to any other cell, the cell reference will still be looking at that particular cell reference. If the absolute cell reference is \$A\$1, the formula will always look at cell A1. It is as if the formula is locked on the cell A1 throughout copy action.

iii. Mixed Cell References with Row Locked – Example: A\$1

- Dollar sign before row reference only.
- Remains absolute or locked when copying across the rows, vertically (up and down).
- Moves relatively when copying across the columns, horizontally (side to side).

iv. Mixed Cell References with Column Locked – Example: \$A1

- Dollar sign before column reference only.
- Remains absolute or locked when copying across the columns, horizontally (side to side).
- Moves relatively when copying across the rows, vertically (up and down).

4) Keyboard to Toggle Cell References = F4 Key.

- i. F4 key = If cursor is touching a cell reference in a formula while in edit mode, F4 toggles between the four basic types of cell references.

4. Excel Functions to Make Calculations Based on one Condition, four examples:

5. SUMIFS Function with One Condition shown on next page in cell H13

6. SUMIFS Function with One Condition & Dynamic Array shown on next page in cell K13

7. COUNTIFS Function with One Condition shown on next page in cell H24

8. COUNTIFS Function with One Condition & Dynamic Array shown on next page in cell K24

	A	B	C	D	E	F	G	H	I	J	K	L
8		Date	Sales Rep	Autos	Sales		Summary Report - Adding with 1 Condition					
9		2/11/19	Alma	Chevy	\$43,000							
10		2/11/19	Tyrone	Chevy	\$71,182		Spreadsheet Formulas:			Office 365 Dynamic Array Formulas:		
11		2/11/19	Jaeyoung	Honda	\$70,720							
12		2/11/19	Alma	Chevy	\$69,598		Autos	Total Sales		Autos	Total Sales	
13		2/11/19	Tyrone	Honda	\$55,299		Chevy	\$270,684		Chevy	\$270,684.00	
14		2/11/19	Jaeyoung	Toyota	\$63,882		Ford	\$230,985		Ford	\$230,985.00	
15		2/12/19	Alma	Toyota	\$47,352		Honda	\$404,296		Honda	\$404,296.00	
16		2/12/19	Tyrone	Ford	\$53,105		Subaru	\$39,500		Subaru	\$39,500.00	
17		2/12/19	Jaeyoung	Toyota	\$32,058		Toyota	\$300,315		Toyota	\$300,315.00	
18		2/12/19	Alma	Honda	\$33,415							
19		2/12/19	Tyrone	Honda	\$71,880		Summary Report - Counting with 1 Condition					
20		2/13/19	Jaeyoung	Ford	\$63,092							
21		2/13/19	Alma	Ford	\$37,432		Spreadsheet Formulas:			Office 365 Dynamic Array Formulas:		
22		2/13/19	Tyrone	Ford	\$45,376							
23		2/13/19	Jaeyoung	Chevy	\$57,034		Autos	Count Sales		Autos	Count Sales	
24		2/13/19	Tyrone	Honda	\$45,881		Chevy	5		Chevy	5	
25		2/13/19	Tyrone	Honda	\$39,950		Ford	5		Ford	5	
26		1/14/19	Tyrone	Subaru	\$39,500		Honda	8		Honda	8	
27		1/14/19	Alma	Ford	\$31,980		Subaru	1		Subaru	1	
28		1/14/19	Tyrone	Toyota	\$45,789		Toyota	6		Toyota	6	
29		1/14/19	Jaeyoung	Honda	\$55,299							
30		1/14/19	Alma	Toyota	\$63,882		Formula in cell H13: =SUMIFS(\$E\$9:\$E\$33,\$D\$9:\$D\$33,G13)					
31		1/14/19	Tyrone	Toyota	\$47,352		Formula in cell K13: =SUMIFS(E9:E33,D9:D33,J13#)					
32		1/14/19	Jaeyoung	Chevy	\$29,870		Formula in cell H24: =COUNTIFS(\$D\$9:\$D\$33,G24)					
33		1/14/19	Alma	Honda	\$31,852		Formula in cell K24: =COUNTIFS(D9:D33,J24#)					

9. Excel Functions to Make Calculations Based on an AND Logical Text.

- i. Functions that can perform AND Logical Tests:
 1. SUMIFS: Adds numbers with one or more conditions or criteria
 2. COUNTIFS: Counts with one or more conditions or criteria
 3. AVERAGEIFS: Averages numbers with one or more conditions or criteria
 4. MAXIFS: Finds Maximum number with one or more conditions or criteria
 5. MINIFS: Finds Minimum number with one or more conditions or criteria
- ii. Each function has arguments that you must enter to make the AND Logical Test Calculation.
- iii. Examples of arguments for SUMIFS and COUNTIFS:

SUMIFS(sum_range, criteria_range1, criteria1, criteria_range2, criteria2...)

- The SUMIFS function adds with 1 or more conditions/criteria.
- **sum_range** argument will contain the range with the numbers.
- **criteria_range1** argument will contain the range with all the items being considered.
- **criteria1** argument contains the specific condition or criteria that tells the function what to consider.
- You can up to 127 pairs of criteria_rangeN and criteriaN arguments

COUNTIFS(criteria_range1, criteria1, criteria_range2, criteria2...)

- The COUNTIFS function counts with 1 or more conditions/criteria.
- **criteria_range1** argument will contain the range with all the items being considered.
- **criteria1** argument contains the condition or criteria that tells the function what to count.
- You can up to 127 pairs of criteria_rangeN and criteriaN arguments

- iv. Examples on Next Page:

10. SUMIFS for Summary Report Based on Adding, Old School Cell References and Dynamic Arrays.

11. COUNTIFS for Summary Report Based on Counting, Old School Cell References and Dynamic Arrays.

12. AVERAGEIFS for Summary Report Based on Averaging, Old School Cell References and Dynamic Arrays.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
4	Goals:															
5	1	We want a Cross Tabulated Report showing Total Sales by Auto and Sales Rep.														
6	2	We want a Cross Tabulated Report showing Average Sales by Auto and Sales Rep.														
7	3	We want a Cross Tabulated Report showing the Count of Transactions by Auto and Sales Rep.														
8	4	We are entering data each day and we want reporting solution to update instantly when we add new sales data.														
9																
10		Date	Sales Rep	Autos	Sales		Cross Tabulated Report for Total Sales (\$)						Cross Tabulated Report for Total Sales (\$)			
11		2/11/19	Alma	Chevy	\$38,016		Autos/ Sales Rep	Alma	Jaeyoung	Tyrone			Autos/ Sales Rep	Alma	Jaeyoung	Tyrone
12		2/11/19	Tyrone	Chevy	\$71,182		Chevy	107,614	57,034	71,182			Chevy	107,614	57,034	71,182
13		2/11/19	Jaeyoung	Honda	\$70,720		Ford	37,432	63,092	98,481			Ford	37,432	63,092	98,481
14		2/11/19	Alma	Chevy	\$69,598		Honda	33,415	70,720	213,010			Honda	33,415	70,720	213,010
15		2/11/19	Tyrone	Honda	\$55,299		Toyota	47,352	95,940	0			Toyota	47,352	95,940	0
16		2/11/19	Jaeyoung	Toyota	\$63,882											
17		2/12/19	Alma	Toyota	\$47,352											
18		2/12/19	Tyrone	Ford	\$53,105		Cross Tabulated Report for Average Sales (\$)						Cross Tabulated Report for Average Sales (\$)			
19		2/12/19	Jaeyoung	Toyota	\$32,058		Autos/ Sales Rep	Alma	Jaeyoung	Tyrone			Autos/ Sales Rep	Alma	Jaeyoung	Tyrone
20		2/12/19	Alma	Honda	\$33,415		Chevy	53,807	57,034	71,182			Chevy	53,807	57,034	71,182
21		2/12/19	Tyrone	Honda	\$71,880		Ford	37,432	63,092	49,241			Ford	37,432	63,092	49,241
22		2/13/19	Jaeyoung	Ford	\$63,092		Honda	33,415	70,720	53,253			Honda	33,415	70,720	53,253
23		2/13/19	Alma	Ford	\$37,432		Toyota	47,352	47,970	0			Toyota	47,352	47,970	0
24		2/13/19	Tyrone	Ford	\$45,376											
25		2/13/19	Jaeyoung	Chevy	\$57,034		Cross Tabulated Report for Count of Transactions						Cross Tabulated Report for Count of Transactions			
26		2/13/19	Tyrone	Honda	\$45,881		Autos/ Sales Rep	Alma	Jaeyoung	Tyrone			Autos/ Sales Rep	Alma	Jaeyoung	Tyrone
27		2/13/19	Tyrone	Honda	\$39,950		Chevy	2	1	1			Chevy	2	1	1
28							Ford	1	1	2			Ford	1	1	2
29							Honda	1	1	4			Honda	1	1	4
30							Toyota	1	2	0			Toyota	1	2	0
31																
32		Old School Cell Reference Formulas:											New School Dynamic Array Formulas:			
33		Formula in cell H12: =SUMIFS(\$E\$11:\$E\$27,\$C\$11:\$C\$27,H\$11,\$D\$11:\$D\$27,\$G12)											Formula in cell H13: =SUMIFS(\$E\$11:\$E\$27,\$C\$11:\$C\$27,H\$11,\$D\$11:\$D\$27,\$G13)			
34		Formula in cell H20: =IFERROR(AVERAGEIFS(\$E\$11:\$E\$27,\$D\$11:\$D\$27,\$G20,\$C\$11:\$C\$27,H\$19),0)											Formula in cell N20: =IFERROR(AVERAGEIFS(E11:E27,D11:D27,M20#,C11:C27,N19#),0)			
35		Formula in cell H27: =COUNTIFS(\$D\$11:\$D\$27,\$G27,\$C\$11:\$C\$27,H\$26)											Formula in cell N27: =COUNTIFS(D11:D27,M27#,C11:C27,N26#)			

13. Screen Shots of Old School Formulas with Totals on Worksheet Named "Old School with Totals".

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Objective #5: SUMIFS for Summary Report Based on Adding										
3												
4		Goals:										
5	1	We want a Cross Tabulated Report showing Total Sales by Auto and Sales Rep.										
6	2	We want a Cross Tabulated Report showing Average Sales by Auto and Sales Rep.										
7	3	We want a Cross Tabulated Report showing the Count of Transactions by Auto and Sales Rep.										
8	4	We are entering data each day and we want reporting solution to update instantly when we add new sales data.										
9												
10		Formula in Cell H15: =SUMIFS(\$E\$14:\$E\$30,\$D\$14:\$D\$30,\$G15,\$C\$14:\$C\$30,H\$14) copied through range H15:J18										
11		The SUM Function is used to calculated the Totals (Alt + = is keyboard for SUM Function).										
12												
13		Date	Sales Rep	Autos	Sales		Cross Tabulated Report for Total Sales (\$)					
14		2/11/19	Alma	Chevy	\$38,016		Autos/ Sales Rep	Alma	Jaeyoung	Tyrone	Total	
15		2/11/19	Tyrone	Chevy	\$71,182		Chevy	107,614	57,034	71,182	235,830	
16		2/11/19	Jaeyoung	Honda	\$70,720		Ford	37,432	63,092	98,481	199,005	
17		2/11/19	Alma	Chevy	\$69,598		Honda	33,415	70,720	213,010	317,145	
18		2/11/19	Tyrone	Honda	\$55,299		Toyota	47,352	95,940	0	143,292	
19		2/11/19	Jaeyoung	Toyota	\$63,882		Total	225,813	286,786	382,673	895,272	
20		2/12/19	Alma	Toyota	\$47,352							
21		2/12/19	Tyrone	Ford	\$53,105							
22		2/12/19	Jaeyoung	Toyota	\$32,058							
23		2/12/19	Alma	Honda	\$33,415							
24		2/12/19	Tyrone	Honda	\$71,880							
25		2/13/19	Jaeyoung	Ford	\$63,092							
26		2/13/19	Alma	Ford	\$37,432							
27		2/13/19	Tyrone	Ford	\$45,376							
28		2/13/19	Jaeyoung	Chevy	\$57,034							
29		2/13/19	Tyrone	Honda	\$45,881							
30		2/13/19	Tyrone	Honda	\$39,950							

SUMIFS in cell H15 adds the Sales for the Sales Rep "Alma" and the Auto "Chevy". In order for a number to be included in the calculation, the record must show the Sales Rep "Alma" and the Auto "Chevy".

Each cell in the interior of this Cross Tabulated Report (Cross-Tab) runs an AND Logical Test calculation to add Sales based on Sales Rep and Auto.

	M	N	O	P	Q	R	S	T	U	V	W
8	Objective #6: COUNTIFS for Summary Report Based on Counting										
9											
10	Formula in Cell N15: =COUNTIFS(\$D\$14:\$D\$30,\$M15,\$C\$14:\$C\$30,N\$14) copied through range N15:P18										
11	The SUM Function is used to calculated the Totals (Alt + = is keyboard for SUM Function).										
12											
13	Cross Tabulated Report for Counting Transactions										
14	Autos/ Sales Rep	Alma	Jaeyoung	Tyrone	Total						
15	Chevy	2	1	1	4						
16	Ford	1	1	2	4						
17	Honda	1	1	4	6						
18	Toyota	1	2	0	3						
19	Total	5	5	7	17						
20											
21											
22	Objective #7: AVERAGEIFS for Summary Report Based on Averaging										
23											
24	Using the AVERAGEIFS function we calculated 4 different averaging formulas to create this report										
25											
26	Cross Tabulated Report for Averaging Transactional Sales (\$)										
27	Autos/ Sales Rep	Alma	Jaeyoung	Tyrone	Total						
28	Chevy	53,807	57,034	71,182	58,958						
29	Ford	37,432	63,092	49,241	49,751						
30	Honda	33,415	70,720	53,253	52,858						
31	Toyota	47,352	47,970		47,764						
32	Total	45,163	57,357	54,668	52,663						
33											
34	Formula in Cell N28: =IF(N15=0,"",AVERAGEIFS(\$E\$14:\$E\$30,\$D\$14:\$D\$30,\$M28,\$C\$14:\$C\$30,N\$27)) copied through range N28:P31										
35	Formula in Cell N32: =AVERAGEIFS(\$E\$14:\$E\$30,\$C\$14:\$C\$30,N27) copied through range N32:P32										
36	Formula in Cell Q28: =AVERAGEIFS(\$E\$14:\$E\$30,\$D\$14:\$D\$30,M28) copied through range Q28:Q31										
37	Formula in Cell Q32: =AVERAGE(E14:E30)										

14. Office 365 Dynamic Array Notes from Worksheet named "O365 Dynamic Arrays".

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1															
2		Objective #8: Office 365 Dynamic Array Functions for Summary Report													
3															
4		In Office 365 Insider Edition, and then at a later release date in 2019, Microsoft has introduced a new Excel Calculation Engine and a new set of special Array Functions.													
5		Some of the new Array Functions are SORT and UNIQUE. TRANSPOSE is an old Array Function that takes a vertical range and converts it to a horizontal range, and Vice Versa.													
6		In addition, if we make a Function Argument Array Operation (putting more than one item into a function argument with the goal of getting the Function to spill multiple answers),													
7		the new Excel Calculation Engine will automatically spill the results in the correct number of cells.													
8		To refer to a Spilled Array, use the top cell in the Spilled Array with a pound sign #, like: H20# to refer to the Spilled Array that lives in cell H20.													
9		For a Spilled Array, the formula lives in the first cell and the remaining cells are gray to indicate that they spilled through the cells but not actually in the cell.													
10		The Dynamic Array Functions will spill automatically and if the source data changes, the function will automatically update.													
11															
12		Goals:													
13		1 We want a Cross Tabulated Report showing Total Sales by Auto and Sales Rep.													
14		2 We are using Office 365 Insider Addition, or later in 2019 Microsoft says that these new Dynamic Array Formulas will be available.													
15		3 We are entering data each day and we want reporting solution to update instantly when we add new sales data.													
16															
17															
18															
19		Date	Sales Rep	Autos	Sales		Cross Tabulated Sales (\$)								
20		2/11/19	Alma	Chevy	\$38,016		Sales Rep / Autos	Alma	Jaeyoung	Tyrone					
21		2/11/19	Tyrone	Chevy	\$71,182		Chevy	\$107,614.00	\$57,034.00	\$71,182.00					
22		2/11/19	Jaeyoung	Honda	\$70,720		Ford	\$37,432.00	\$63,092.00	\$98,481.00					
23		2/11/19	Alma	Chevy	\$69,598		Honda	\$33,415.00	\$70,720.00	\$213,010.00					
24		2/11/19	Tyrone	Honda	\$55,299		Toyota	\$47,352.00	\$95,940.00	\$0.00					
25		2/11/19	Jaeyoung	Toyota	\$63,882										
26		2/12/19	Alma	Toyota	\$47,352										
27		2/12/19	Tyrone	Ford	\$53,105		Formula in Cell G21: =SORT(UNIQUE(D20:D36))								
28		2/12/19	Jaeyoung	Toyota	\$32,058		Formula in Cell H20: =TRANSPOSE(SORT(UNIQUE(C20:C36)))								
29		2/12/19	Alma	Honda	\$33,415		Formula in Cell H21: =SUMIFS(E20:E36,C20:C36,H20#,D20:D36,G21#)								
30		2/12/19	Tyrone	Honda	\$71,880										
31		2/13/19	Jaeyoung	Ford	\$63,092										
32		2/13/19	Alma	Ford	\$37,432										
33		2/13/19	Tyrone	Ford	\$45,376										
34		2/13/19	Jaeyoung	Chevy	\$57,034										
35		2/13/19	Tyrone	Honda	\$45,881										
36		2/13/19	Tyrone	Honda	\$39,950										